

Dans ce TP, on reprend les exemples de lissage exponentiel proposés sous SAS.

1 Lissage de Holt-Winters

On s'intéresse aux ventes automobiles mensuelles aux Etats-Unis. La table de données `usecon` est disponible sous SAS dans la librairie `sashelp`.

1. Tracer la série temporelle et donner ses principales caractéristiques.

```
symbol1 i=spline v=dot;
axis2 label=(a=-90 r=90 "Vehicles and Parts" )
       order=(6000 to 24000 by 3000);

title1 "Sales of Passenger Cars";
proc sgplot data=sashelp.usecon;
    series x=date y=vehicles / markers;
    xaxis values=('1jan80'd to '1jan92'd by year);
    yaxis values=(6000 to 24000 by 3000);
    format date year4.;
run;
```

2. Prédire les ventes automobiles mensuelles par un lissage de Holt-Winter à l'horizon 12 mois.

```
proc forecast data=sashelp.usecon interval=month
    method=winters seasons=month lead=12
    out=out outfull outresid outest=est;
    id date;
    var vehicles;
```

```

    where date >= '1jan80'd;
run;

```

- (a) Utiliser l'aide en ligne pour bien comprendre chaque option de l'appel de la PROC FORECAST.
- (b) Interprétez les résultats.
- (c) Tracer et commenter les résidus.

```

title1 "Sales of Passenger Cars";
title2 'Plot of Residuals';
proc sgplot data=out;
    where _type_ = 'RESIDUAL';
    needle x=date y=vehicles / markers markerattrs=(symbol=circlefilled);
    xaxis values=('1jan80'd to '1jan92'd by year);
    format date year4.;
run;

```

- (d) Tracer la série temporelle des prévisions ainsi que l'intervalle de confiance associé.

```

title1 "Sales of Passenger Cars";
title2 'Plot of Forecast from WINTERS Method';
proc sgplot data=out;
    series x=date y=vehicles / group=_type_ lineattrs=(pattern=1);
    where _type_ ^= 'RESIDUAL';
    refline '15dec91'd / axis=x;
    yaxis values=(9000 to 25000 by 1000);
    xaxis values=('1jan90'd to '1jan93'd by qtr);
run;

```

2 Lissage exponentiel double

On cherche maintenant à prévoir les ventes de pétrole et produits dérivés aux Etats-Unis.

1. Tracer la série temporelle et la décrire.

```

title1 "Sales of Petroleum and Related Products";
proc sgplot data=sashelp.usecon;
    series x=date y=petrol / markers;
    xaxis values=('1jan80'd to '1jan92'd by year);
    yaxis values=(8000 to 20000 by 1000);

```

```
format date year4.;
run;
```

2. Prédire les ventes de pétrole par un lissage exponentiel double à l'horizon 12 mois.

```
proc forecast data=sashelp.usecon interval=month
              method=expo trend=2 lead=12
              out=out outfull outest=est;
  id date;
  var petrol;
  where date >= '1jan80'd;
run;
```

3. Interprétez les résultats.
4. Tracer la série temporelle des prévisions ainsi que l'intervalle de confiance associé.

```
title1 "Sales of Petroleum and Related Products";
title2 'Plot of Forecast: EXPO Method';
proc sgplot data=out;
  series x=date y=petrol / group=_type_;
  xaxis values=('1jan89'd to '1jan93'd by qtr);
  yaxis values=(10000 to 20000 by 1000);
  refline '15dec91'd / axis=x;
run;
```